2.4. Использование подпрограмм и модулей

	Пример 2.4.1. Создать процедуру PutCar(x,y,m), которая рисует машину с координатами центра (x,y) и масшт. коэффициентом  m.

Нарисовать несколько машин.



	
	Uses Graph;

Var gd, gm, k: Integer; 

Procedure PutCar(x,y,m: Integer); 

begin
{ Процедура отрисовки машины }

   SetFillStyle(1,14); Bar(x-3*m,y-m,x+m,y+m);

   SetFillStyle(1,12); Bar(x+m,y-2*m,x+3*m,y+m);

   SetFillStyle(1,7); FillEllipse(x-2*m,y+m,m,m);

   FillEllipse(x+2*m,y+m,m,m);

end;  {PutCar}

begin {Main}

    gd:=Detect; InitGraph(gd,gm,'');

    PutCar(100,100,10);  PutCar(300,200,7);  

    For k:=1 To 5 Do  PutCar(k*100,300,5);  

    Readln; CloseGraph;

end.


	Пример 2.4.2. Машина движется по экрану слева направо.


	Uses Graph, Crt;

Var gd, gm: Integer;  

        x, y, m: Integer; 

Procedure PutCar(x,y,m: Integer); 

begin


   {... см.пример. 2.4.1. }

end;  {PutCar}

Procedure DelCar(x,y,m: Integer);

begin

   SetFillStyle(1,0); 

   Bar(x-3*m,y-2*m,x+3*m,y+2*m); 

end;  {DelCar}

begin {Main}

   gd:=Detect; InitGraph(gd,gm,'');

   x:=100; y:=200; m:=10;


   PutCar(x,y,m); Delay(1000);

   Repeat 
     DelCar(x,y,m);

     x:=x+2; 

     PutCar(x,y,m); Delay(1000); 

   Until (x>=500) or KeyPressed;


   CloseGraph;

end.
	Graph - для рисования, Crt - для KeyPressed

координаты машины и масштаб

PutCar рисует машину 

DelCar стирает машину черным прямоугольником

Начальные координаты машины

Delay зависит от быстродействия PC

В цикле x меняется от 100 до 500 с шагом 20 

     стереть машину со старого места

     изменить координаты

     нарисовать машину на новом месте

конец цикла, если x(500 или нажата клавиша



Большую программу можно разделить на несколько файлов - модулей. Обычно в каждом модуле - несколько подпрограмм. Стандартные модули мы использовали, начиная с первого занятия (Crt для оформления экрана в текстовом режиме и позже Graph для рисования).

А сейчас попробуем создать свой модуль.

	Пример 2.4.3. Создать модуль Pictures.pas, содержащий поцедуры отрисовки простых объектов: дом, дерево, человечек и т.п.  Создать программу Pic.pas, которая рисует пейзаж, используя процедуры модуля Pictures.


	
 Модуль Pictures.pas:
	Программа Pic.pas:

	Unit Pictures;

INTERFACE    {раздел интерфейса - заголовки}

Procedure House(x,y:Word);

Procedure Car(x,y:Word);

…

IMPLEMENTATION   { раздел реализации }

Uses Graph;

Procedure House (x,y: Integer);

begin

   ...  {операторы - отрисовка дома }

end;  {House}

Procedure Car(x,y: Integer);

begin

   ... { операторы - отрисовка машины }

end;  {Car}

...  

begin   

end.
	Uses Graph, Pictures;

var gd,gm: Integer;

      x: Integer;

begin

    gd:=Detect; InitGraph(gd,gm,'');

    House (100,100);

    Car(200,200); 

    For x:=1 To 5 Do

        Tree(x*70,350);

    Readln; CloseGraph;

end.


Обратите внимание: модуль *.pas компилируется в файл *.tpu,  который можно использовать в любой другой программе так же, как и стандартные модули Graph, Crt и другие.

2.5. 
Анализ событий клавиатуры

Коды нажатых клавиш сначала попадают в специальный блок памяти - буфер клавиатуры. Затем программа извлекает их из буфера и обрабатывает.

Модуль CRT предоставляет следующие средства для анализа буфера клавиатуры:

	If KeyPressed ... - если была нажата клавиша... (если в буфере есть символы)

ReadKey - читать символ из буфера клавиатуры


	Пример 2.5.1. Программа читает символы из буфера клавиатуры и выводит на экран их коды
	
	Uses Crt;

Var k1, k2: Char;

begin

   ClrScr; Writeln('Press Button!');

   Repeat

     k1:=#0; k2:=#0;

     If KeyPressed Then

       begin

        k1:=ReadKey;

        If k1=#0 Then k2:=ReadKey;

        Writeln('kodi: ',Ord(k1),' - ',Ord(k2));

       end;

   Until k1=#27;

end.

	
	

	Цикл до нажатия <Esc> 

   если никакая клавиша  не нажата - #0

   если нажата клавиша,

      читаем из буфера 

        соответствующие символы k1 и k2

        и выводим коды этих символов

конец цикла  (27 - код <Esc>)
	


	Пример 2.5.2. "Пианино"


	
	Uses Crt;

Var k1, k2: Char;

begin

   Repeat

     k1:=#0; k2:=#0;

     If KeyPressed Then

       begin

          k1:=ReadKey; If k1=#0 Then k2:=ReadKey;

          Case k2 of {F1-do, F2-re, ... - 1.okt.}

              #59: Sound(131);

              #60: Sound(147);

          end; {Case}

          Case k1 of {Q-do, W-re, ... - 2.okt.}

              'Q','q': Sound(262);

              'W','w': Sound(294);

          end; {Case}

         Delay(1000); NoSound;

       end;  {If}

   Until k1=#27;

end.

	
	
	

	Цикл до <Esc> 

   клавиши не нажимались - #0 

   Если нажата клавиша,

        читаем из буфера k1 и k2

       первая октава:

            <F1> - до

            <F2> - ре

             ...

       вторая октава:

            <Q> - до

            <W> - ре

             ...

конец цикла (27 - код <Esc>)
	


	Пример 2.5.3. Машина движется вправо-влево под управлением клавиш-стрелок.
	
	Пример 2.5.4. При нажатии любой клавиши включается звук; если отпустить клавишу, звук выключается. Выход - <Esc>.

	
	
	

	Uses Graph, Crt;

Const m=15; step=20;

Var  x, y, gd, gm: Integer;

        k1, k2:     Char;

Procedure PutCar(x,y,m: Integer);

begin



   { нарисовать машину - прим.. 2.4.1.}

end;  {PutCar}

Procedure DelCar(x,y,m: Integer); 


begin


     { стереть машину - прим.. 2.4.2.}

end;  {DelCar}

begin {Main}

   gd:=Detect; InitGraph(gd,gm,'');

   x:=200; y:=200; PutCar(x,y,m);

   Repeat { цикл до <Esc> }

     k1:=#0; k2:=#0; 

     If KeyPressed Then          {если нажата}

       begin            {клавиша, читать символ} 

          k1:=ReadKey;                 {из буфера}
          If k1=#0 Then k2:=ReadKey;
          Case k2 of         { движение }

             #75: begin      { Left - налево }

                        DelCar(x,y,m);

                        If  x>step+3*m Then 

                             x:=x-step;

                        PutCar(x,y,m);

                     end;

             #77: begin     { Right - направо }

                        DelCar(x,y,m);

                        If  x<640-step-3*m Then

                                   x:=x+step;

                        PutCar(x,y,m);

                     end;

           end; { Case }

        end; { If KeyPressed }

   Until (k1=#27); { <Esc> }

   CloseGraph;

end.
	
	

	
	
	

	
	
	Чтобы засечь момент отпускания клавиши, в цикле читаем 96-ой элемент из порта ввода (p:=Port[96]) и ждем, когда его значение изменится. Высота звука в этой программе соответствует номеру клавиши - это 98-ой элемент из порта ввода.  

	
	
	Uses  Crt;

var p,s: LongInt;

begin

   Repeat

         p:=Port[96];  

         s:=Port[98];

         If s in [1..101] Then

                           Sound(s*50);

         Repeat

             { ждем }

         Until (Port[96]<>p);

         NoSound;

   Until s=1;

end.

	
	
	Для получения правильных звуков (соответствующих клавишам пианино), в процедуру Sound  надо передать более сложную формулу, например:

Sound(Round(110*exp(s*ln(2)/12)));



2.6. Анализ событий мышки

Прерывания - это стандартные функции OS, обрабатывающие события, например, нажатие клавиши или перемещение мышки. Для вызова функции прерывания надо:

- подключить модуль DOS и описать регистровую переменную (типа Registers);

- задать поля регистровой переменной, которые используются в обработке прерывания (AX содержит номер функции, BX, CX, DX - данные);  

- вызвать процедуру обработки прерывания:

     Intr(номер_прерывания, регистровая_переменная); 

- выходные данные можно прочитать из полей регистровой перменной.

Для обработки событий мышки можно использовать функции прерывания (33)16:

AX=1 - сделать видимым (показать) курсор мышки;



AX=2 - спрятать курсор мышки;

AX=3 - определить координаты мышки и состояние кнопок:

на выходе:
BX = кнопки: 

0 - ни одна не нажата





1 - нажата левая кнопка





2 - нажата правая кнопка


CX = горизонтальная координата,


DX = вертикальная координата;

AX=4 - установить курсор мышки в точку (x,y):

на входе:
CX=x,
DX=y;

AX=7 - задать диапазон активности мышки в горизонтальном направлении:

на входе:
СX=минимальный x, DX=максимальный x;

AX=8 - задать диапазон активности мышки в вертикальном направлении:

на входе:
СX= минимальный y, DX= максимальный y;

	Пример 2.6.1. Если нажата левая кнопка мышки, программа рисует точку; по правой кнопке - выход из программы


	Uses Dos, Graph;

var  gd,gm: Integer;

       r: registers;

begin    gd:=Detect; InitGraph(gd,gm,'');

  r.ax:=1; intr($33,r); 

  r.ax:=4; r.cx:=320; r.dx:=240; intr($33,r);   

  Rectangle(200,100,400,300);

  r.ax:=7; r.cx:=200; r.dx:=400; intr($33,r); 

  r.ax:=8; r.cx:=100; r.dx:=300; intr($33,r);

  Repeat     

       r.ax:=3; intr($33,r);        

       If r.bx=1 Then    
            

              PutPixel(r.cx, r.dx-1, 15);     

   Until r.bx=2;
   CloseGraph;

end.
	регистровая переменная 

показать курсор 

поставить курсор в точку (320,200) 

ограничить движение мышки 

прямоугольником (200,100)-(400,300)

цикл до нажатия правой кнопки 

   определить положение мышки

   если нажата левая кнопка, нарисовать

       точку на 1 выше курсораа

конец цикла (нажата правая кнопка)


2.7. Палитра 

Pascal позволяет изменять стандартную цветовую палитру (0-черный, 1-синий, 2-зеленый и т.д.), меняя номера цветов и даже сами цвета.

	SetPalette(c,n) - согласовать цвет c и номер цвета n

SetRGBPalette(c,r,g,b) - создать цвет c - смесь r частей красного (red), g - зеленого (green) и b - синего (blue); r,g,b принимают значения от 0 до 63.


	Пример 2.7.1. Змея с ярко-зеленой головой и более темным телом бегает по кривой, заданной параметрическими уравнениями: 

	Uses Graph, CRT;

Var  gd,gm, xe,ye,c: Integer;  

       t: Real;

Const   step=0.05; m=50;   

            xc=320; yc=240;

 begin

    gd:=Detect; InitGraph(gd,gm,'');

    For c:=0 To 15 Do    

      begin                      

         SetPalette(c,c);   SetRGBPalette(c, 0, 3+c*4, 0);

      end;

    t:=0;    

    Repeat   
       For c:=0 To 15 Do  

          begin  

             SetFillStyle(1,c); SetColor(c);

             FillEllipse(Round(xc+m*cos(t+c*step)),   

                    Round(yc-m*sin(3*(t+c*step))),3,3);

          end;

       t:=t+step; Delay(1000);
    Until KeyPressed; 
    CloseGraph;

end.
	создать палитру: 

 15 зеленых оттенков 

  и один черный (нулевой)

в цикле меняется t (время)

   тело змеи -  15 кругов,

   каждый своего оттенка 

t+c*step: c-тый сегмент обгоняет хвост на c шагов

(c=0 - хвост, c=15 - голова)

Delay зависит от быстродействия PC 




2.8. 
Динамическая память в графическом режиме

Динамическая память ("куча", heap) используется для временного хранения данных, особенно если их размер заранее не известен. Прежде чем "положить" данные в динамическую память, надо инициализировать указатель (GetMem), а после использования данных необходимо освободить память (FreeMem). Некорректное использование динамической памяти может привести к "зависанию" программы.

ImageSize(x1,y1,x2,y2) - объем памяти, который занимает рисунок, расположенный 

                                           в прямоугольнике (x1,y1,x2,y2)
P: Pointer
- указатель на переменную не определенного заранее типа (адрес)

P^ 

- данные, хранящиеся по адресу Р
GetMem(P, v);
- выделить для хранения данных участок памяти объема v 

                               и определить значение P - его адрес

FreeMem(P, v);
- освободить память по адресу P в объеме v
GetImage(x1,y1,x2,y2,p^)       - сохранить прямоугольный рисунок 

                                      в динамической памяти по адресу P

PutImage(x,y, p^, NormalPut) - вывести на экран рисунок начиная с (x,y) из

                                                     динамической памяти по адресу P в нормальном виде

(NormalPut, OrPut, XorPut, AndPut, NotPut - способы взаимодействия с фоном)

Пример 2.8.1. Колобок перемещается по цветному фону, не затирая его.
	Uses Graph, Crt;

Var  gd, gm, IS: Integer;

        p: Pointer;

        x,y: Word;

Procedure ColorBack; { цветной фон }

var i: Word;

begin

   For i:=1 To 15 Do

    begin

     SetFillStyle(1,i); SetColor(i);

     FillEllipse(320,240,320-20*i, 240-15*i);

    end;

end;  { ColorBack}

Procedure Kolobok(x,y: Word); {колобок}

begin

   SetFillStyle(1,14); SetColor(14); 

   FillEllipse(x,y,40,40);

   SetFillStyle(1,1); SetColor(1);

   FillEllipse(x-10,y,5,5);   

   FillEllipse(x+10,y,5,5);

   SetFillStyle(1,12); SetColor(12);

   FillEllipse(x,y+20,15,5);

end;  {Kolobok}
	begin {Main}

    gd:=Detect; InitGraph(gd,gm,'');

    ColorBack;

    x:=100; y:=100;

    { IS - размер изображения Колобка }

    IS:= ImageSize(x-50,y-50,x+50,y+50);
    {выделить память для хранения фона}

    GetMem(P,IS);            {под Колобком}

    Repeat
       {запомнить фон}

       GetImage(x-50,y-50,x+50,y+50,p^);
       {нарисовать Колобка}

       Kolobok(x,y);   Delay(2000);
       {восстановить фон}

       PutImage(x-50,y-50, p^, NormalPut);
       {изменить координаты}

       x:=x+20; y:=y+15;

    Until KeyPressed or (x>600) or (y>440);
    {освободить память}

    FreeMem(P, IS);    CloseGraph;

end.

