2. Работа в графическом режиме

2.1. Модуль Graph: некоторые функции и процедуры

Инициализация/конец работы:

gd:=Detect; InitGraph(gd,gm,'') - инициализация

       графического режима;

CloseGraph - выход из графического режима;

Не забудьте скопировать в свою рапку файл 

       egavga.bgi (из TP6 или из BP\BGI).

Фигуры:

PutPixel(x,y,c) - точка (x,y) цвета c;

Line(x1,y1,x2,y2) - отрезок, соединяющий (x1,y1) и (x2,y2);

MoveTo(x1,y1) - поставить карандаш в точку (x1,y1);

LineTo(x2,y2) - провести отрезок в точку  (x2,y2);

Rectangle(x1,y1,x2,y2) - прямоугольник с диагонилью (x1,y1)-(x2,y2);

Bar(x1,y1,x2,y2) - закрашенный прямоугольник;

Bar3D(x1,y1,x2,y2,d,True) - параллелепипед (d-глубина);

FillEllipse(x,y,Rx,Ry) - закрашенный овал с центром (x,y) и полуосями (Rx,Ry);

[image: image8.bmp][image: image9.bmp]Circle(x,y,R) - окружность с центром (x,y) и радиусом R;

Ellipse(x,y,a1,a2,Rx,Ry) - дуга овала от a1( до  a2(;

Sector(x,y,a1,a2,Rx,Ry) - сектор овала;

Стили:

SetColor(c) - цвет карандаша (линии) c;

SetFillStyle(s,c) - стиль заливки s (1-сплошная), цвет заливки c;

SetBkColor(c) - цвет фона c;

ClearDevice - очистка экрана цветом фона;

FloodFill(x,y,c) - заливка замкнутой области вокруг точки (x,y) до границы цвета c;

Текст:

STR(x,S) - преобразование числа x в строку S;

OutTextXY(x,y,S) - вывод на экран строки S; координаты начала строки (x,y);

SetTextStyle(f, d, s) - стиль текста: f - фонт, d - направление, s - размер;

Перед началом работы скопируйте в свою папку файл egavga.bgi.

	Пример 2.1.1. Овалы и прямоугольники.
	

	
	

	Uses Graph;               {подключить модуль Graph}

var gd,gm: Integer;   {граф. устройство и режим }

begin                          

   gd:=Detect; InitGraph(gd,gm,'');

   SetColor(1); SetFillStyle(1,14);  {цвета}

   Circle(100,100,25);                     {окружность}

   FillEllipse(100,200,25,50);   {закрашенный овал}

   SetColor(2); SetFillStyle(1,12);  {смена цвета}

   Rectangle(200,100,250,150);     {прямоугольник}

   Bar(200,200,300,250);          {закраш. прямоуг. }

   Readln; CloseGraph;

end.
	

	
	a)


                   b)


         c)



	Пример 2.1.2. Круги в циклах.
	
	

	
	

	{a} For r:=1 To 5 Do  Circle(100, 100, 10*r);

{b} For x:=1 To 5 Do  Circle(80*x, 100, 30);

{c} For x:=1 To 5 Do  

           For y:=1 To 4 Do

               Circle(80*x, 80*y, 30);
	

	
	

	Пример 2.1.3. Квадраты.
	
	

	
	

	Uses Graph;

var gd,gm: Integer;

      k: Word;   { номер квадрата }

begin

   gd:=Detect; InitGraph(gd,gm,'');

   For k:=1 To 5 Do

     begin

        if k mod 2 = 0 Then  {четные-контуры}

           Rectangle(50*k,50*k,50*k+30,50*k+30)

        Else   {нечетные-закрашенные}
           Bar(50*k,50*k,50*k+30,50*k+30);

     end;

   Readln; CloseGraph;

end.
	Четные k - Rectangle,

Нечетные k - Bar;

Размеры: 30 - длина стороны,

                  50 - "шаг"


2.2. 
Графики кривых

Декартова система координат

	Пример 2.2.1. Построить график функции y=x2-5 на отрезке [-4, 4].
	
	Uses Graph;

Const

   xc=320; yc=240; {координаты центра}

   m=15;                  {масштаб}

   step=0.01;            {шаг}

Var

   x,y: Real;             {матем. координаты}

   xe,ye: Integer;      {экранные координаты }

   gm,gd: Integer; 

begin     

   gd:=Detect; InitGraph(gd,gm,''); 

   Line(xc,10,xc,470); {оси}

   Line(10,yc,630,yc);

   x:=-4;     {цикл: точки на графике}

   Repeat
       y:=x*x-5;

       xe:=Round(xc+m*x); {Round - округлить}

       ye:=Round(yc-m*y);  {до целых}

       PutPixel(xe,ye,2);

       x:=x+step;

    Until x>4;
    Readln; CloseGraph;

end.

	              
	
	

	               Необходимо учесть:

1) экранные координаты точки (XE,YE) не совпадают с математическими (X,Y):

               XE=XC+X,

               YE=YC-Y;

2) масштаб рисунка M:


XE=XC+M*X,


YE=YC-M*Y;

	Строя графики, надо учитывть область определения функции и "обходить" точки, в которых знаменатель равен нулю или почти нулю: 

If abs(знаменатель)>0.01 Then  поставить точку; {иначе ничего не делать}


	Пример 2.2.2. Построить график функции  y=tg(4x) для x ( [-(, (].
	
	...

   x:=-pi; {цикл: точки на графике}

   Repeat
       If  abs(cos(4*x))>0.01 Then

          begin

              y:=sin(4*x)/cos(4*x);

              xe:=Round(xc+m*x); 

              ye:=Round(yc-m*y);  

              PutPixel(xe,ye,14);

          end;

       x:=x+step;

    Until x>pi;
...

	
	
	



Полярная система координат

	Если кривая задана уравнением 

 в полярной системе координат, то в цикле меняется (, по данной формуле вычисляется (; зная ( и (, можно найти декартовы координаты точки по формулам: 

EMBED Equation 

	Uses Graph;

Const   xc=320; yc=240; m=10;  step=0.01; 

Var      x,y,r,fi: Real;     

            xe,ye: Integer; 

            gm,gd: Integer; 

 Begin          

   gd:=Detect; InitGraph(gd,gm,''); 

   Line(xc,10,xc,470);   Line(10,yc,630,yc);

   fi:=0; 

   Repeat

         r:=fi;  

         x:=r*cos(fi); 

         y:=r*sin(fi);

         xe:=Round(xc+m*x);

         ye:=Round(yc-m*y);

         PutPixel(xe,ye,15);

         fi:=fi+step;
    Until fi>6*pi;
    Readln; CloseGraph;

End.

	
	
	

	Пример 2.2.3. Построить спираль (3 витка). Уравнение спирали в полярных координатах: (=(
(( меняется от 0 до 6().
	
	

	
	
	


Параметрические уравнения

	В цикле меняется параметр  t (время?); x и y вычисляются как функции t.
	
	Uses Graph;

var  x,y,t: Real;

       xe,ye: Integer;

       gd,gm: Integer;

Const  step=0.01; m=80;  xc=320; yc=240;

 begin

   gd:=Detect; InitGraph(gd,gm,'');

   Line(xc,10,xc,470);   Line(10,yc,630,yc);
   t:=0;

   Repeat

       x:=cos(t);

       y:=sin(3*t);
       xe:=Round(xc+m*x);

       ye:=Round(yc-m*y);

       PutPixel(xe,ye,14);

       t:=t+step;

   Until t>2*Pi;

   Readln; CloseGraph;

end.

	
	
	

	Пример 2.2.4. Построить кривую, заданную параметрическими уравнениями:

Рисунки на основе кривых

	 Пример 2.2.5. Нарисовать:
	
	Uses Graph;

var  x,y,t: Real;

     xe,ye,n: Integer;

     gd,gm: Integer;

Const   step=0.01; m=30;  xc=320; yc=340;

 begin

   gd:=Detect; InitGraph(gd,gm,'');

   x:=-2*pi;  n:=0;

   Repeat

       y:=sin(x);

       xe:=Round(xc+m*x);

       ye:=Round(yc-m*y);

       PutPixel(xe,ye,2);

       inc(n);

       If n mod 10 =0 Then

        begin

          SetColor(12);Line(150,150,xe,ye);

          SetColor(14);Line(350,150,xe,ye);

        end;

       x:=x+step;

   Until x>2*Pi;

   Readln;   CloseGraph;

end.



	
	
	

	n - номер точки

на каждую десятую точку

опустить линию из "фонарей" - точек (150,150) и (350,150)
	
	


2.3. 
Использование случайных чисел в графическом режиме

	Randomize - инициализация генератора случайных чисел

Random(N) - случайное целое число в диапазоне  [0..N) = [0..N-1]

M+Random(N) - случайное целое число в диапазоне [M..M+N) = [M..M+N-1]


	Пример 2.3.1. Заполнить экран точками, координаты и цвет которых определяют случайные числа.
	
	Пример 2.3.2. "Муха на стекле". Со временем координаты точки (мухи) меняются случайным образом на (5 по горизонтали или по вертикали.

	
	
	

	...

Randomize;
Repeat { цикл до нажатия клвиши }

    x:=20+Random(601);  {x: 20..620}

    y:=20+Random(441);  {y: 20..460}

    c:=1+Random(15);      {цвет: 1..15}

    PutPixel(x,y,c);

    Delay(1000);

Until KeyPressed;

...
	
	...

   Randomize;
   x:=320; y:=240;

   PutPixel(x,y,15); Delay(1000);

   Repeat

          PutPixel(x,y,0); 

          napr:=Random(4);   {направление}       

          Case napr of

             0: If x<620 Then x:=x+5;  {(}

             1: If x>20 Then x:=x-5;     {(}

             2: If y<460 Then y:=y+5;  {(}

             3: If y>20 Then y:=y-5;     {(}

          end;

          PutPixel(x,y,2); Delay(1000);

   Until KeyPressed;

...

	
	
	


	Пример 2.3.3. В графическом режиме создать тест, содержащий задачи с параметрами - случайными числами.


	     …

ball:=0;      

{задача 1}

SetFillStyle(1,1); Bar(0,0,639,479);    { синее окно для задачи}

SetColor(14);


x:=5+Random(6); Str(x,xx);   { x - число [5..10],  хх - соответствующая строка }

y:=5+Random(6); Str(y,yy);   { y - число [5..10],  yy - соответствующая строка }

OutTextXY(50,100,' 1. Из зоопарка сбежало ' + xx + ' змей ' );

OutTextXY(50,130, 'по '+yy + ' метров каждая. ' );

OutTextXY(50,160,' Сколько метров змей сбежало из зоопарка? ');

Readln(z);  

If z=x*y Then inc(ball);          { если ответ верен, увеличивается ball }

{задача 2}

     ...

