4. Массивы

4.1. Одномерные массивы

Массивы предназначены для хранения однотипных данных. Массивы тесно связаны с циклами: элементы массива и запрашивают, и выводят, и обрабатывают в цикле.

	Описать переменные:
     имя_массива: array[диапазон_изменения_индекса] of тип_элементов

Var   A: array[1..10] of Word;   - в массиве A может быть до10 натуральных чисел

         С: array[0..99] of Char;     - в массиве С может быть до 100 символов 

Описать константы (точнее, инициализированные переменные):

     имя_массива: array[диапазон_изменения_индекса] of тип_элементов =

                                                                                                               (значения);

Const S: array[1..5] of String = ('Anna', 'Alla', 'Ausma ', 'Arta', 'Anete ');

           R: array[1..3] of Real = (3.1415, 2.17, 9.87);

Обратиться к отдельным элементам:

     имя_массива [индекс]

  С[1] - первый элемент массива C
      С[0] - нулевой элемент массива C 

  А[5] - пятый элемент массива A                     A[k] -тый элемент массива A

	Запросить массив X, содержащий 

N целых чисел:

Write('N='); Readln(N);

For k:=1 To N Do

  begin

     Write('X[',k,']='); Readln(X[k]);

  end;
	Вывести на экран мавссив Y, содержащий N действительных чисел:

For k:=1 To N Do

    Writeln('Y[',k,']=',Y[k]:8:2);


	Piemērs 4.1.1. Используя массивы вопросов и правильных ответов (Q и A), создать тест по арифметике.


	Uses CRT;

Const N=5;    

         Q: Array[1..N] of String= ('2*8=', '7*4=', '4*5=', '3*6=', '7*3=');

         A: Array[1..N] of Word= (16, 28, 20, 18, 21);

Var   k, ball, x: Word;
	кол-во вопросов

вопросы
прав. ответы



	begin

     ClrScr;   ball:=0;

     For k:=1 To N Do

        begin

            Write(Q[k]); Readln(x); 

            If x=A[k] Then inc(ball);

        end;

     Writeln('Result=',ball); Readln;

end.
	До теста кол-во правильных ответов ball=0

В цикле программа:

  выводит очередной вопрос Q, запрашивает x,

  если ответ x совпадает с правильным, растет ball

Обратите внимание: размер тела программы не зависит от количества вопросов!


	Пример 4.1.2. Программа запрашивает массив A, содержащий N целых чисел, и вычисляет:

a) сумму и среднее арифметическое для всех элементов

b) количество тех чисел, которые больше, чем стеднее арифметическое,

c) самое маленькое среди всех чисел,

d) самое большое среди четных чисел.


	Uses CRT;

var   k, N, b, min, sum, Np, max: Integer;   

        av: Real;

       A: Array[1..100] of  Integer;

begin

    ClrScr; Write('N='); Readln(N);   

     For k:=1 To N Do

         begin

             Write('A[',k,']='); Readln(A[k]);

          end;

{a}sum:=0;

     For k:=1 To N Do sum:=sum+A[k];

     av:=sum/N; 

     Writeln('sum=', sum, '   av=', av:8:2);

{b}b:=0;

     For k:=1 To N Do

          If  A[k]>av Then inc(b);

     Writeln('>av : ',b);

{c} min:=A[1]; 

       For k:=2 To N Do

           If A[k]<min Then min:=A[k];

     Writeln('min= ',min);

{d} Np:=0;

     For k:=1 To N Do

         If A[k] mod 2 =0 Then 

            begin

              inc(Np); 

              If Np=1 Then 

                   max:=A[k]

              Else If A[k]>max Then 

                   max:=A[k];

            end;

     If Np=0 Then Writeln('четных нет')

     Else Writeln('max pāra sk. = ',max);

     Readln;

end.
	В массиве A может быть до 100 целых чисел (т.к. вряд ли учитель введет больше)

Запросить N и элементы массива

Числа сохранены в массиве, поэтому можно делать каждый пункт отдельно 

sum - сумма всех элементов

av - среднее арифметическое

b - количествово чисел, больших среднего

Надеемся, что первый - самый маленький

В цикле проверяем остальные элементы:

    если есть меньший, min меняется

Np - количество четных чисел

В цикле:

    Если A[k] четное число, то:

         Np увеличивается на 1,

         Первое встретившееся четное число

             считаем наибольшим

         иначе проверяем: если очередное 

             число больше, max меняется

Прежде чем выводить max, надо проверить, были ли вообще четные числа.




Примечание. Хорошо бы проверить, что введеное значение N не превышает 100.

4.2. 
Создание массивов по формулам и по условию

	Пример 4.2.1. С помощью генератора случайных чисел создать массив A, содержащий N целых чисел в диапазоне [-10; 10] и вывести его на экран. Каких чисел в массиве больше: отрицательных или положительных?


	Uses CRT;

var   k, N, Nn, Np: Integer;   

       A: Array[1..100] of  Integer;

begin

    ClrScr; Randomize;

    Write('N='); Readln(N); 

    Write('A:  ');  

     For k:=1 To N Do

         begin

             A[k]:=-10+Random(21);

             Write(A[k],'   ');

          end; 

    Nn:=0; Np:=0; Writeln;

    For k:=1 To N Do

          If A[k]>0 Then inc(Np)

          Else If A[k]<0 Then inc(Nn); 

     Writeln('Np=', Np, '  Nn=', Nn);

     If Np=Nn Then Writeln('+ = -')

     Else If Np>Nn Then Writeln('+ > -')

     Else Writeln('+ < -');

     Readln;

end.
	В массиве A может быть до 100 целых чисел

Инициализация генератора случ. чисел

Запросить N - количество чисел в массиве

В цикле создать и вывести массив A

A[k] - случайное число [-10; 10] 

(в этом диапазоне 21 число, считая 0)

Nn - количество отрицательных чисел,

Np - количество положительных чисел

После цикла - вывод результатов


	Пример 4.2.2. Создать массив A, содержащий N целых чисел - квадраты первых N натуральных чисел (1, 4, 9, ...)


	Uses CRT;

var   k, N: Integer;   

       A: Array[1..100] of LongInt;

begin

    ClrScr;

    Write('N='); Readln(N); 

    Write('A:  ');  

     For k:=1 To N Do

         begin

             A[k]:=k*k;

             Writeln('A[',k,']=', A[k]);

          end;

     Readln;

end.
	В массиве A  может быть до100 целых чисел

Запросить N - количество чисел в массиве

В цикле создать и вывести массив



	Пример 4.2.3. Программа запрашивает массив A, содержащий N целых чисел, и:

a) создает массив X по формулам: X1=A1; X2=A1+A2; X3=A1+A2+A3; ... XN=A1+...+AN;

b) создает массив Y, содержащий все четные числа из массива A.


	Uses CRT;

var   k, N, Ny: Integer;   

       A, X, Y: Array[1..100] of Integer;

begin

    ClrScr; Write('N='); Readln(N);   

     For k:=1 To N Do

         begin

             Write('A[',k,']='); Readln(A[k]);

          end;

{a}X[1]:=A[1]; Writeln('X[', 1, ']=', X[1]);

     For k:=2 To N Do

         begin

                X[k]:=X[k-1]+A[k];

                Writeln('X[', k, ']=', X[k]);

         end;

{b}Ny:=0;

     For k:=1 To N Do

        If A[k] mod 2 = 0 Then

           begin

                inc(Ny);

                Y[Ny]:=A[k];

                Writeln('Y[', Ny, ']=', Y[Ny]);

           end;

     If Ny=0 Then Writeln('Y is empty'); 

     Readln;

end.
	В массивах м.б. до 100 целых чисел

Запросить N и массив

X1=A1; 

X2=A1+A2 = X1+A2;

X3=A1+A2+A3 = X2+A3; ...

XN=A1+...+AN = XN-1+AN.

Ny - кол-во четных чисел в массиве A

Если A[k] - четное число, то:

    Ny увеличивается на 1

    A[k] записывается в массив Y

Если в массиве A четных чисел не было, Y остается пустым


4.3. 
Массивы строк

	Var S: Array[1..10] of String; - в массиве S может быть до 10 строк

                                                  (в каждой - до 255 символов)

Const B: Array[1..3] of String = ('Abc', 'Defg', 'Klmnop');

B[i][k] - k-тый символ i-той строки массива B, напр.: B[1][2]='b', B[2][1]='D'.


	Пример 4.3.1. Программа запрашивает массив S, содержащий N строк. a) сколько из них длиннее, чем последняя?

b) в скольки из них есть одинаковые символы (напр., 'asds', 'df3g3r')?


	Uses CRT;

var   i, j, k, N, x, y,t: Integer;   

       S: Array[1..20] of String;

begin

    ClrScr;

    Write('N='); Readln(N); 

     For k:=1 To N Do

         begin

             Write('S[',k,']='); Readln(S[k]);

          end;

{a}x:=0;

     For k:=1 To N-1 Do

         If Length(S[k])> Length(S[N]) Then

            inc(x);

     Writeln('a) A[k]>A[N] : ', x);

{b}y:=0;

     For i:=1 To N Do

        begin

            t:=0;

            For k:=1 To Length(S[i])-1 Do 

               For j:=k+1 To Length(S[i]) Do

                  If S[i][k]=S[i][j] Then inc(t);

            If  t>0 Then inc(y);

        end;

     Writeln('b) y=', y);  Readln;

end.
	В массиве м.б. до 20 строк

Запросить N и массив

x - количество строк, которые длиннее

     последней

y - количество строк, в которых есть 

      одинаковые символы

В цикле проверяем i-тую строку

t - кол-во совпадений в i-той строке

каждый символ, кроме последнего,

   сравниваем со всеми последующими;

      если символы одинаковы, растет t

   если в i-той строке есть одинаковые   

   символы, растет y

Выводим результат




4.4. 
Двухмерные массивы

	Var A: Array[1..3, 1..4] of Integer; - в двухмерном массиве (в таблице) A может быть до 3x4 целых чисел

A[2,3] - элемент массива A с "координатами" (2,3) 

Двухмерные массивы обрабатывают в двойном цикле; например, вывести таблицу:

For x:=1 To 3 Do  For y:=1 To 4 Do   

   begin  

        GotoXY(x*7,y); Write(A[x,y]; 

   end;


	Пример 4.4.1. Программа запрашивает двухмерный массив А - таблицу целых чисел МxN - и создает одномерный массив В, содержащий средние арифметические значения, вычисленные для строк массива A.


	Uses Crt;

var A: Array[1..10,1..10] of Integer;

      B: Array[1..10] of Real;

      N, M, x, y: Word;

begin 

    ClrScr;   

    Repeat

       Write('Кол-во строк = ');  Readln(N);

       Write('Кол-во столбцов= ');  Readln(M);

    Until (M in[1..10]) and (N in [1..10]);

    GotoXY(3, 5); Writeln('Введи A:');

    For x:=1 To M Do For y:=1 To N Do

       begin

          GotoXY(5*x, 5+y); Readln(A[x,y]);

       end;

    GotoXY(63,5); Write('B:');  

    For y:=1 To N Do  

      begin

          B[y]:=0;  

          For x:=1 To M Do  B[y]:=B[y]+A[x,y];

          B[y]:=B[y]/M;  

          GotoXY(60,5+y); Write(B[y]:6:2);

      end;

    Readln;

end.
	Нарпимер:

A:                B:

2  4  6           4.00

3  7  5           5.00

1  2  1           1.33

Надо проверить, что 1(M,N(10

При вводе элементов массива A

курсор занимает соответствующую позицию в таблице

B[y]  - среднее арифм. значение элементов строки y массива A

сначала находим сумму элементов

потом - среднее арифм. значение

B выводит справа от A





	Пример 4.4.2. Программа рисует маленькими кружочками и квадратиками букву А, шаблон которой хранится в массиве A. 


	Uses Crt,Graph;

Const N=5; M=6;  

      A: Array[1..N,1..M] of Byte=

     ((0,0,1,1,1,1),  

      (0,1,0,1,0,0),    

      (1,0,0,1,0,0),    

      (0,1,0,1,0,0),

      (0,0,1,1,1,1));

var   gd,gm: Integer;

        x,y: Word;

begin

   gd:=Detect; InitGraph(gd,gm,'');

   For x:=1 To N Do      

       For y:=1 To M Do  

           If A[x,y]=1 Then  

             Circle(50+5*x,50+5*y,2);

   For x:=1 To N Do   

       For y:=1 To M Do

          If A[x,y]=1 Then

              Bar(100+5*x,50+5*y,

                     100+5*x+3,50+5*y+3);

   Readln; CloseGraph;

end.
	Обратите внимание: шаблон лежит на боку, так как на самом деле в двухмерном массиве первый индекс - это номер строки, а не координата x, как нам хотелось бы.

N - количество строк, M - кол-во столбцов 

A - буква А, составленная из единиц 

A из кружочков:

     если A[x,y]=1, рисуем кружочек

     если A[x,y]=0, не рисуем ничего 

А из квадратиков:

     5 - расстояние между центрами квадратов,

     3 - длина стороны квадрата


Двухмерные массивы часто используются в компьютерных играх (шахматная доска, лабиринты, поля для игр "морской бой", "тетрис" и т.п.). 

В математике и физике одномерный массив соответствует вектору; с помощью двухмерных массивов описывают таблицы, матрицы, тензоры, графы.

В языке Pascal, как и во многих других языках программирования, можно использовать также массивы больших размерностей (трехмерные и т.п.)

4.5. 
Сортировка массивов

Поиск и сортировка, по-видимому, самые популярные задачи в программировании. Применяя различные алгоритмы сортировки к большим массивам, можно заметить весьма значительную разницу в их эффективности (до в 100 раз). Но если массив маленький, можно смело использовать самый простой алгоритм, не переживая по поводу его неэффективности.

	Пример 4.5.1. В массивах Lat и Rus хранятся существительные на латышском и русском языках соответственно. Программа умеет переводить их и сортировать "словарь" в алфавитном порядке.


	Uses Crt;

Const  N=5;   

     Lat: Array[1..N] of String= ('suns', 'govs', 

                                        'gailis', 'vista', 'aita');

     Rus: Array[1..N] of String=('собака', 'корова',

                                        'петух', 'курица', 'овца');

Var   L, R: String;         

          i, k: Word;

begin

    ClrScr;      

    Write('Ievadi vārdu / введи слово: '); Readln(L);

    k:=0; 

    For i:=1 To N Do

          If (L=Lat[i]) OR (L=Rus[i]) Then

              begin

 
     inc(k); Writeln(Lat[i],'  -  ',Rus[i]);

             end;

    If k=0 Then Writeln(' ??? ');

    Writeln('Pēc kārtošanas / после сортировки:');   

    For i:=1 To N-1 Do           

      For k:=i+1 To N Do                

          If Lat[k]<Lat[i] Then    

             begin                          

                 L:=Lat[i]; Lat[i]:=Lat[k]; Lat[k]:=L;

                 R:=Rus[i]; Rus[i]:=Rus[k]; Rus[k]:=R;

             end;

    For i:=1 To N Do Writeln(Lat[i],'  -  ',Rus[i]); 

    Readln;

end.
	N - количество слов в словаре

поиск и перевод слова 
k - количество найденных слов 

В цикле ищем слово L:

  Если L = i-тому слову в словаре,

      k увеличивается на 1;

      выводим слово и его перевод

Если L в словаре нет, выводим '?'

сортировка словаря по алфавиту 

каждый элемент Lat[i] сравниваем 

  со всеми последующими Lat[k]; 

    если следующий "меньше" 

    предыдущего, 

        меняем их местами 

        (латышские и русские слова

          надо менять одновременно)

Выводим отсортированный словарь

