3. Строки

3.1. Действия с отдельными символами

	Описание переменных и констант:
Var 
s: String; 
- строка, длина которой не превышает 255 символов,


s1: String[10];
- строка, длина которой не превышает 10 символов.

Const     Name='Svetlana';

Длина строки S (количество символов в строке S):     Length(S)

Обращение к отдельным символам:

S[1]               - первый символ строки S,             S[3] - третий символ строки S,

S[Length(S)] - последний символ строки S,       S[k] - k-тый символ строки S.

Добавление-замена символов:

Добавить в конец строки S восклицательный знак:

S:=S+'!';

Добавить в начало строки S три вопросительных знака: 
S:='???'+S;

Третий символ строки S заменить на звездочку:
               S[3]:='*';


	Пример 3.1.1. a) Чему равна длина введенной строки?

                         b) Чему равен первый символ строки?

                         c) Сколько символов 'V' в строке?

                         d) Есть ли в строке хотя бы один символ 'V'?

                         e) Состоит ли строка только из символов 'V'?


	Uses Crt;

Var S: String;

       x, k: Integer;

begin

   ClrScr;

   Write('S='); Readln(S);

   {a} Writeln('длина S =', Length(S));

   {b} Writeln('S[1]=',S[1]);

   {c} x:=0;

   For k:=1 To Length(S) Do   

          If S[k]='V' Then 

                inc(x);

  Writeln('x=',x);

  {d} If  x>0 Then Writeln('Есть V')

         Else Writeln('Нет V');

  {e} If x=Length(S) Then Writeln('Только V')

         Else Writeln('Не только V'); 

   Readln;

end.
	S - строка

x - количествово символов 'V' и  

k - переменная цикла - целые числа

Запросить строку S 

Длина строки

Первый символ

x - количество символов 'V' 

В цикле просматриваем каждый символ S[k]; если это 'V', 

    то x увеличивается на 1

Если x>0, то 'V' в строке есть

Если x = длина строки (Length),

то строка состоит только из 'V'.





	Пример 3.1.2. Во введенной строке заменить все вопросительные знаки на восклицательные.


	Uses Crt;

Var S: String;     k: Integer;

begin

   ClrScr;   Write('S=');  Readln(S);

   For k:=1 To Length(S) Do

        If S[k]='?' Then S[k]:='!';

   Writeln('new S=', S); Readln;

end.
	В цикле просматирваем все символы строки S 

      если k-тый символ '?', то присваиваем ему 

                                   новое значение '!'


	Для строк определены операции сравнения (сравниваются коды символов), например, 'Alina'>'Aleksandra' (коды латинских букв растут в алфавитном порядке).

Если k-тый символ строки S - заглавная латинская буква…:


If (S[k]>='A') AND (S[k]<='Z') Then…       или       If S[k] in ['A'..'Z'] Then…

Если k-тый символ строки S - цифра…:


If (S[k]>='0') AND (S[k]<='9') Then…        или        If S[k] in ['0'..'9'] Then…


	Пример 3.1.3. Каких символов в строке S больше: заглавных или строчных латинских букв?


	{фрагмент программы}

x1:=0; x2:=0;

For k:=1 To Length(S) Do       

    If  S[k] in ['A'..'Z']  Then inc(x1)      

    Else If  S[k] in ['a'..'z']  Then inc(x2);

If x1>x2 Then Writeln('больше заглавных')

Else If x2>x1 Then Writeln('больше строчных')

Else Writeln('заглавных и строчных поровну');
	x1 - количество заглавных букв,

x2 - количество строчных букв 

В цикле просматирваем все символы

    Если S[k] лат. загл., растет x1,

    если лат. стр.,- растет x2

После цикла сравниваем значения x1 и x2 


	Пример 3.1.4. Есть ли в строке S одинаковые символы?


	{ фрагмент программы }

x:=0;

For k:=1 To Length(S)-1 Do       

   For i:=k+1 To Length(S) Do   

      If S[i]=S[k] Then inc(x);

If x>0 Then Writeln('Yes')

Else Writeln('No');
	x - количество повторяющихся символов

Каждый символ, кроме последнего,

   сравниваем со всеми последующими

       Если они одинаковы, x растет

Если x>0, то одинаковые есть; если x=0, то нет.

Примечание. Алгоритм можно улучшить - выйти из циклов после первого же совпадения. 


3.2. 
Действия с фрагментами строки

	Copy(S,k,x)         - подстрока строки S начиная с k-того символа длины x: 

Pos(S1,S)
- позиция подстроки S1 в строке S;

                               (если фрагмента S1 в строке S нет, то Pos(S1,S)=0)
Delete(S,n,x)
- удалить из строки S подстроку начиная с n-ого символа длины x;

Insert(S1,S,n)
- вставить строку S1 в строку S начиная с n-той позиции.


Конкретные примеры помогут понять, как работают эти функции и процедуры.

	Фрагмент программы (S, S1, S2: String; x1,x2,x3: Integer;)
	Результат

	S:='crocodile'; S1:=Copy(S,4,3); S2:=Copy(S,7,2);
	S1='cod', S2='il'

	S:='crocodile'; x1:=Pos('roc',S); x2:= Pos('il',S); x3:= Pos('rod',S);
	x1=2, x2=7, x3=0

	S:='crocodile'; S1:=S;  Delete(S,4,2); Delete(S1,1,7);
	S='crodile', S1='le'

	S:='crocodile'; Insert('roro',S,6);
	S:='crocororodile';


	Пример 3.2.1. Программа запрашивает латышское существительное первого склонения и склоняет его (без проверки).


	Write ('S: '); Readln(S);   

S1:= Copy(S,1,Length(S)-1);     

Writeln('N.  ',S);

Writeln('Ģ.  ',S1+'a');

Writeln('D.  ',S1+'am');

…
	если S='skolotājs', то

S1='skolotāj' - корень слова

N. skolotājs

Ģ. skolotāja

D. skolotājam

...


	Пример 3.2.2. a) Есть ли во введенной строке имя  'Anna'?

                         b) Сколько раз оно встречается?


	Uses Crt;

Var S: String;  x, k: Integer;

begin

  ClrScr;   Write('S='); Readln(S);

 {a} If  Pos('Anna',S)>0 Then Writeln('Yes')

       Else Writeln('No'); 

 {b} x:=0;

  For k:=1 To Length(S)-3 Do

      If  Copy(S,k,4)='Anna' Then  inc(x);

  Writeln('x=', x); Readln;

end.
	Запросить строку

Ответить на первый вопрос очень легко

Ответ на второй вопрос ищем в цикле

S='Anna11Anna22Anna'

      

                     S[13]=S[Length(S)-3]

Примечание. Алгоритм можно улучшить.





	Пример 3.2.3. Во введенной строке заменить все фрагменты 'cat' на 'mouse'.


	Uses Crt;

Var S: String;  p: Integer;

begin

  ClrScr;   Write('S='); Readln(S);

  p:=Pos('cat',S); 

  While p>0 Do

     begin 

        Delete(S,p,3); 

        Insert('mouse',S,p); 

        p:=Pos('cat',S);

     end;

  Writeln('new S=', S); Readln;

end.
	Запросить строку

p - позиция фрагмента 'cat' в строке S

Цикл: пока в строке S есть фрагменты 'cat'

     удалить 'cat' из S (3 символа, начиная с p)

     вставить 'mouse' на то же мместо

     найти следующий фрагмент 'cat'

конец цикла

Примечание. Удалить из S фрагмент S1, начинающийся с p:  Delete(S,p,Length(S1));


	Пример 3.2.4. Запросить строку, в которой записано время в формате hh:mm:ss. Записать его в формате hh ч. mm мин. ss сек.


	Uses Crt;

Var S,S1: String;

begin

  ClrScr;   Write('S='); Readln(S);

  S1:=Copy(S,1,2)+' ч. '+

          Copy(S,4,2)+' мин. '+

          Copy(S,7,2)+' сек. ';

  Writeln('S1=', S1); Readln;

end.
	время:                     03:13:54

номера символов: 12345678

Примечание. Возможны и другие решения.


3.3. 
Преобразование типов

Ord(c)
- порядковый номер (код) символа с;

Chr(x)
- символ, номер (код) которого x;

       Коды некоторых символов:

        'A' - 65, 'B' - 66, 'C' - 67, ..., 'Z' - 90;

         'a' - 97, 'b' - 98, 'c' - 99, ..., 'z' - 122;

        '0' - 48, '1' - 49, '2' - 50, ..., '9' - 57.

Str(x,S)
- преобразование числа x в строку S;

Str(x:8:2,S) - по формату: всего 8 символов, из них 2 после десятичной точки;

Val(S,x,ec) - преобразование строки S в число x; ec - код ошибки 

                   (если преобразование прошло успешно, то ec=0, 

                    иначе ec - это позиция символа, из-за которого преобразование 

      невозможно.

                    Например, после Val('23.5',x,ec) получится: x=23.5 и ec=0;  

                                       после Val('22 km',x,ec) получится: x - неопределено и ec=3;)

	Пример 3.3.1. Программа запрашивает символ и сообщает его код (номер)
	
	Пример 3.3.2. Запрашивается код символа и выводится сам символ

	
	
	

	Uses Crt;

Var c: Char;  {символ}

       x: Byte;  {целое число [0..255]}

begin

      ClrScr; Write('c='); Readln(c);

       x:=Ord(c);

      Writeln('Ord(c)=', Ord(c));    Readln;

end.
	
	Uses Crt;

Var x: Byte;  {целое число [0..255]}

begin

      ClrScr; Write('x='); Readln(x);

     Writeln('Chr(x)=', Chr(x));

     Readln;

end.


	Пример 3.3.3. Закодировать введенное слово, заменив все латинские заглавные буквы кроме Z на следующие в алфавите; Z - на A; остальные символы - на звездочки.


	Uses Crt;

Var S: String;   k: Integer;

begin

   ClrScr; 

   Write('S: '); Readln(S);

   For k:=1 To Length(S) Do

      If S[k]='Z' Then S[k]:='A'

      Else If S[k] in ['A'..'Y'] Then 

           S[k]:=Chr(Ord(S[k])+1)

      Else S[k]:='*';

   Writeln('new S=',S); Readln;

end.
	Ja S[k]='B':   Chr(Ord('B')+1)

                      Chr(     66    +1)

                            Chr(67)         

                                'C'

В цикле просмотреть все символы

   'Z' заменить на 'A'

   буквы от 'A' до 'Y' - на следующие в алфавите

   остальные символы - на звездочки





	Пример 3.3.4. В строке S записано время в формате hh:mm. Проверить корректность ввода и найти количество минут, прошедших от начала суток. 


	Uses Crt;

Var S: String;   h, m, ec1, ec2: Integer;

begin

   ClrScr; 

   Write('S: '); Readln(S);

   Val(Copy(S,1,2),h,ec1);

   Val(Copy(S,4,2),m,ec2);

   If (ec1<>0) Or (ec2<>0) Or (S[3]<>':')

       Or (Length(S)<>5) Then 

          Writeln('Format Error')

   Else If (h<0) Or (h>23) Or (m<0) 

       Or (m>59) Then   Writeln('Time Error')

   Else     Writeln('x=',h*60+m); 

   Readln;

end.
	Пример времени:   ='11:08'

Номера символов:     12345

Copy(S,1,2)='11';   h=11;

Copy(S,4,2)='08';   m=8;

Если ec(0, это значит, что вместо цифр стоят другие символы

h([0..23], m([0..59]

в часе 60 минут


	Пример 3.3.5. Создать строку, содержащую квадраты первых N натуральных чисел (N<=100), например, для N=4: S='14916'.
	
	S:=''; Write('N='); Readln(N);

For k:=1 To N Do

   begin

       Str(k*k,SS); S:=S+SS;

   end;

Writeln('S=',S);

	
	
	


	Необязательные функции и операции

Succ - следующий символ (или число)

Pred - предыдущий символ (или число)

Inc - увеличить значение на 1 (следующий символ)

Dec - уменьшить значение на 1 (предыдущий символ)

UpCase - преобразовать строчную лат. букву в заглавную


	Пример 3.3.6. Программа демонстрирует употребление функций Succ, Prev, UpCase и операций inc, dec.

(c: Char)
	
	Write('Введите строчную лат.букву: '); Readln(c);

Writeln('Предыдущий символ: ', Pred(c));

Writeln('Следующий символ: ', Succ(c));

Writeln('Соотв. заглавная буква: ', UpCase(c));

Writeln('было: c=',c);

inc(c); Writeln(' после inc(c):  c=',c);

dec(c); Writeln(' после dec(c):  c=',c);

	
	
	


3.4. 
Более сложные задачи

	Пример 3.4.1. Программа запрашивает N строк и сообщает:

a) сколько символов в самой длинной строке,

b) сколько строк начинаются с буквы 'S'


	Uses Crt;

Var S: String;   

       max, y, N, k: Integer;

begin

   ClrScr;  y:=0;

   Write('N= '); Readln(N);

   For k:=1 To N Do

     begin

          Write('S= '); Readln(S); 

          {a} If k=1 Then max:=Length(S)

           Else If  Length(S)>max  Then 

                                     max:=Length(S);

          {b} If  S[1]='S' Then inc(y)

     end;

   Writeln('max=',max, '    y=',y); Readln;

end.
	N - количество строк,

x - длина самой длинной строки

y - кол-во строк, начинающихся с 'S' 

k - переменная цикла - номер строки

Запросить количество строк

В цикле запросить и обработать 

              N строк:

   Пусть первая строка - самая длинная 

   Если новая строка длиннее, 

                   max меняет свое значение

   Если S начинается с 'S', растет y


	Пример 3.4.2. Программа запрашивает N строк и сообщает, сколько среди них симметричных (например, 'ALLA'). 


	Uses Crt;

Var S: String;    x, N, k, i: Integer;

       Ok: Boolean;

begin

 ClrScr;  x:=0;

 Write('N= '); Readln(N);

 For k:=1 To N Do

   begin

     Write('S= '); Readln(S); Ok:=true;
      For i:=1 To Length(S) div 2 Do

         If  S[i]<>S[Length(S)-i+1] Then Ok:=false;

      If  Ok  Then inc(x);

   end;

 Writeln('x=',x); Readln;

end.
	N - кол-во строк; k,i - перем. цикла,

x - кол-во симметричных строк,

Ok=true, если строка симметрична

и Ok=false, если нет

Запросить кол-во строк

В цикле: 

    Запросить очередную строку

    Надеемся, что S симметрична   

    Проверяем, так ли это

    Если Ok осталось true, x растет

Пример строки: 'ABCDCXA'

                   i=2      Length(S)-i+1=6





	Пример 3.4.3. В строке символов записана сумма двух натуральных чисел. Вычислить результат (без проверки корректности). Например, '123+45' ( '123+45=168'


	Uses Crt;  {I вариант}

Var  S, SS: String;   x, y, k: Integer;

begin

   ClrScr;  Write('S= '); Readln(S);

   k:=1; x:=0;  {первое число}

   While S[k] in ['0'..'9'] Do 

      begin 

         x:=x*10+Ord(S[k])-48;  inc(k);

      end;

   inc(k); y:=0; {второе число}

   While S[k] in ['0'..'9'] Do 

      begin 

          y:=y*10+Ord(S[k])-48;  inc(k);

      end;

   Str(x+y,SS); S:=S+'='+SS; 

   Writeln(S); Readln;

end.
	Uses Crt;  {II вариант}

Var  S, SS: String;   x, y, ec, p: Integer;

begin

   ClrScr;  Write('S= '); Readln(S);

   p:=Pos('+', S);  

   Val(Copy(S,1,p-1),x,ec);

   Val(Copy(S,p+1,Length(S)-p),y,ec);

   Str(x+y,SS); S:=S+'='+SS; 

   Writeln(S); Readln;

end.


	Пример 3.4.4. В строке S записано некоторое выражение с круглыми скобками. Проверить правильность расстановки скобок. Например: '(a)(()(c))' - да, '((a)))(()' - нет. 


	Uses Crt;  

Var  S, SS: String;   k: Integer;

begin

   ClrScr;  Write('S= '); Readln(S);

   SS:='';

   For k:=1 To Length(S) Do

       If S[k]='(' Then SS:=SS+'('

       Else If S[k]=')' Then

            If Length(SS)>0 Then 

                  Delete(SS,Length(SS),1)

            Else 

                begin

                     Writeln('No'); Readln; Halt;

                end;

   If  SS='' Then Writeln('Yes')

   Else Writeln('No'); Readln;

end.
	Используем строку SS как стек для хранения  уже открытых, но еще не закрытых скобок. (В случае, когда есть только круглые скобки, стек не так уж нужен - достаточно следить за кол-вом  открытых но не закрытых скобок). 

В цикле анализируем каждый символ:

     '(' записываем в стек SS

     встретив ')', выкидываем из стека '(', 

         (если есть что выкидывать)

     если ')' есть, а стек пустой,

     скобки стоят неправильно, напр. (()))...

Если после цикла стек не пустой, то снова

скобки стоят неправильно, напр. ((())()   

